
This lecture

1 2 3 4 5 6 7 8 9 10 11 12 13 14 15

I · what a proof is II · how proofs are built III · correct programs IV · AI in the loop

↓ you are here theory in Lean

Definitions

 ▷ () := M : N;
instantiation (U)

def, theorem,
arguments

▶ A name stands
for a term, and
using the name
is applying it to
arguments.

1

Unfolding

δ-reduction,
δ-normal form,
βδ-conversion

#print, unfold,
rfl

▶ Equality by
computation now
includes
unfolding names,
and the checker
does it silently.

2

Primitives

() := ⊥⊥ : N;
axioms and
primitive notions

axiom, #print
axioms

▶ What a proof
assumes is a
finite list of
names, and the
checker can print
it.

3

Scripts

every line of a flag
proof is a named
definition

have, And.intro,
obtain

▶ A proof is a list
of definitions,
each line named
and usable later,
and the machine
writes them from
the goal.

4

Trust

the de Bruijn
criterion

#print, sorry

▶ Whatever built
the term, only
the term is
checked, and
only the axioms
it names are
assumed.

5

History: every mathematics book is an environment

Euclid
fl. 300 BC

Definitions,
postulates, and

propositions that
cite earlier ones by

number.

c. 300 BC Euclid’s Elements: definitions with a body,
postulates without one, and every propo-
sition built from what came before.

1968 De Bruijn’s Automath: a “book” is a list
of lines, each a definition or a primitive
notion, and a machine checks the book.

1990s Barendregt names the de Bruijn criterion:
whatever builds the term, a small kernel
checks it.

2021 Lean 4: #print axioms lists the primitive
notions a theorem rests on.

N. G. de Bruijn
1918–2012

Definitions as
first-class citizens
of a type theory.

λD makes Euclid’s structure a formal object: an environment of definitions, some with a body and
some without. Photos: Wikimedia.

What a definition is

Definition (definiendum, definiens)
In  := E, the name  is the definiendum, the defined constant; the
expression E is the definiens.

A definition needs a setting: “total” needs a set and a relation on it. The
setting is a context, written in flags.

S : ∗

R : (S × S)→∗

tot(S,R) := ∀,y∈S(R(, y)∨ R(y, )) : ∗

The list (S,R) is the parameter list: the subjects of the context flags, in order.
N&G Def. 8.3.1, Rem. 8.3.2, Fig. 8.1

A series of definitions in one context

() S : ∗

(b) R : (S × S)→∗

(1) reƒ ee(S,R) := ∀∈S(R(, )) : ∗

(2) ntsymmetrc(S,R) := ∀,y∈S((R(, y)∧ R(y, ))⇒  = y) : ∗

(3) trnste(S,R) := ∀,y,z∈S((R(, y)∧ R(y, z))⇒ R(, z)) : ∗

(4) prty-ordered(S,R) :=

reƒ ee(S,R)∧ (ntsymmetrc(S,R)∧ trnste(S,R)) : ∗

Left of :=, (S,R) is a parameter list; in line (4) the same lists are
instantiations. Hence the order: (4) after (1) to (3). N&G §8.5, Fig. 8.2

The series, in Lean

-- The two flags are the two binders; the type follows ‘:‘, the definiens ‘:=‘.
def Reflexive (S : Type) (R : S → S → Prop) : Prop := ∀ x, R x x
def Antisymm (S : Type) (R : S → S → Prop) : Prop :=
∀ x y, R x y → R y x → x = y

def Transitive (S : Type) (R : S → S → Prop) : Prop :=
∀ x y z, R x y → R y z → R x z

def PartialOrder (S : Type) (R : S → S → Prop) : Prop :=
Reflexive S R ∧ Antisymm S R ∧ Transitive S R -- (4) after (1)-(3)

#check @Reflexive -- @Reflexive : (S : Type) → (S → S → Prop) → Prop

Outside its definition the constant takes the flags as arguments: #check shows the
parameter list as a -type.

Using a definition: instantiation

A constant is born in its definition, then lives by instantiation of its
parameters.

first flag second flag

S : ∗ R : (S × S)→∗

instantiation R : ∗ ≤ : (R× R)→∗
instantiation N+ : ∗ | : (N+ × N+)→∗

Instantiating S changes the type of R: in the first row ≤ is checked against
((S × S)→∗)[S := R], the flag’s type after the earlier substitution.
tot(N+ , |) is a correct instantiation, and false. N&G §8.4, Rem. 8.4.1

Instantiation is application

-- Two instantiations; the type of R is computed from the choice of S.
#check Reflexive Nat (· ≤ ·) -- Reflexive Nat (fun x y => x ≤ y) : Prop
#check Reflexive String (· ̸= ·) -- Reflexive String (fun x y => x ̸= y) : Prop

-- Under S := Nat, a relation on strings is refused:
#guard_msgs(drop error) in
#check Reflexive Nat (fun (s t : String) => s = t)
-- Application type mismatch: the argument has type String → String → Prop
-- but is expected to have type Nat → Nat → Prop

-- A correct instantiation, a false proposition, and so refutable:
example : Reflexive Nat (· < ·) → False := fun h => Nat.lt_irrefl 0 (h 0)

Flags in Lean: variable and section

-- ‘variable‘ raises the flag; ‘section ... end‘ is its pole.
section
variable (S : Type) (R : S → S → Prop)

def Refl : Prop := ∀ x, R x x -- parameters unwritten
def Trans : Prop := ∀ x y z, R x y → R y z → R x z
def Preord : Prop := Refl S R ∧ Trans S R -- and written when used
end -- the flag is hauled down

#check @Preord -- Preord : (S : Type) → (S → S → Prop) → Prop

Four kinds of definition

Write ∗s for ∗ read as “sets” and ∗p for ∗ read as “propositions”; the subscript
is a reading aid, not a new sort.  ▷ (. . . ) := E : T writes the flags  on one line,
before the definition they govern.

(sets) 1 ▷ A(. . . ) := E1 : ∗s
(objects) 2 ▷ (. . . ) := E2 : S with S : ∗s
(propositions) 3 ▷ B(. . . ) := E3 : ∗p

(proofs) 4 ▷ b(. . . ) := E4 : P with P : ∗p

The fourth names a proof: under PAT, a term like any other. It is what makes a
theorem applicable.
The one asymmetry: a notion is defined in order to be unfolded; a proof is used through
its type, and its body is not unfolded. N&G Not. 8.7.1, Figs. 8.4–8.5, §16.1

A theorem is a definition

-- ‘theorem‘ for proofs, ‘def‘ for the other three kinds.
def Nat2 : Type := Nat -- sets
def two : Nat := 2 -- objects
def Even (n : Nat) : Prop := ∃ k, n = 2*k -- propositions
theorem even4 : Even 4 := 〈2, rfl〉 -- proofs

#print two -- def ...two : Nat :=
-- 2

#print even4 -- theorem ...even4 : Even 4 :=
-- Exists.intro 2 rfl

Giving names to proofs

Bézout’s Lemma: if m,n are coprime, gcd(m,n) = 1, there are integers , y
with m+ ny = 1. The third flag is an assumption: a proof, declared like any
other variable.

() m : N+

(b) n : N+

(c)  : coprme(m,n)

p(m,n, ) := formalproof : ∃, y:Z. (m + ny = 1)

Without the name, using the lemma at 55,28 means copying the proof with 55
and 28 substituted throughout. With it, p(55,28, U) is the specialised proof, for
any U : coprme(55,28).
A theorem is applied by instantiation. N&G §8.6–8.8, Figs. 8.3, 8.6–8.7

A theorem applied by instantiation, in Lean

-- 1. A theorem with three data parameters and one proof parameter, u.
theorem le_add_right_of_le (a b k : Nat) (u : a ≤ b) : a + k ≤ b + k :=

Nat.add_le_add_right u k

-- 2. Instantiate the data: the type of u becomes 3 ≤ 5.
#check le_add_right_of_le 3 5 10 -- 3 ≤ 5 → 3 + 10 ≤ 5 + 10

-- 3. Supply U : 3 ≤ 5 as the last argument: the specialised theorem.
example : 3 + 10 ≤ 5 + 10 := le_add_right_of_le 3 5 10 (by decide)

This is p(55,28, U): instantiate the data parameters, then supply the proof parameter.

The environment Δ

Definition
(environment)
An environment Δ is a
finite list of definitions; an
extended judgement
has the form Δ ;  ⊢ M : N.
Δ binds the constants in
M : N,  the free variables.

N&G Def. 9.1.2, 9.2.1, Rem. 9.2.2, 9.2.4

Two rules: adding a definition, and using one

(deƒ ) — adding a definition

Δ ;  ⊢ K : L Δ ; :A ⊢ M : N

Δ,D ;  ⊢ K : L
( fresh in Δ) with D ≡ :A ▷ () := M : N

:A are the parameter flags 1:A1, . . . , n:An; U below are the arguments supplied for them.

(nst) — using one

Δ ;  ⊢ ∗ : □ Δ ;  ⊢ U : A[ := U]

Δ ;  ⊢ (U) : N[ := U]
()

The first premiss says only that Δ and  are well-formed; it also covers an empty parameter list.
Instance. (deƒ ) adds reƒ ee to Δ from Δ ; S:∗, R:(S × S)→∗ ⊢ ∀∈S(R(, )) : ∗; then (nst)
gives reƒ ee(N,≤) : ∗.
Not an instance. reƒ ee(N,=R): the second premiss asks for U2 : (N× N)→∗, and =R has type
(R× R)→∗. N&G Def. 9.3.1, 9.4.7, §9.4

A file is an environment

-- A file is Δ; ‘variable‘ and a declaration’s binders are .
variable (S : Type)
def Reflexive (R : S → S → Prop) : Prop := ∀ x, R x x
def PartialOrder (R : S → S → Prop) : Prop := Reflexive S R

#print PartialOrder
-- def ...PartialOrder : (S : Type) → (S → S → Prop) → Prop :=
-- fun S R => Reflexive S R
--  has become λ.

δ-reduction unfolds one occurrence

Definition (one-step δ-reduction)
If  ▷ () := M : N is in Δ, then (U)→δ M[ := U], also inside any larger
expression.

With n:N ▷ d(n) := n+ n : N in Δ:

One occurrence at a time; the other d stays folded. N&G Def. 9.5.1, Rem. 9.5.3

rw takes one δ-step

def d (n : Nat) : Nat := n + n -- n:N ▷ d(n) := n + n : N, now in Δ

-- ‘rw [d]‘ unfolds one occurrence of d: the first it finds, here d a.
example (a b : Nat) : d a + d b = (a + a) + (b + b) := by

rw [d] -- ⊢ a + a + d b = a + a + (b + b) d b is still folded
rw [d] -- ⊢ a + a + (b + b) = a + a + (b + b) both sides agree: closed

δ-normal form: nothing left to unfold

Definition (δ-reduction, δ-conversion, δ-normal form)
M↠δ N: zero or more δ-steps. M =δ N: steps in either direction, a backward
step being a folding. K is in δ-normal form when no constant with a body
occurs in it.

N&G Def. 9.5.4, 9.5.5, 9.5.7

unfold takes every δ-step for one constant

-- ‘unfold d‘ unfolds every occurrence of d: the δ-normal form with respect to d.
example (a b : Nat) : d a + d b = (a + a) + (b + b) := by

unfold d -- ⊢ a + a + (b + b) = a + a + (b + b)
rfl

-- With respect to d only: a constant defined through d stays folded.
def e (n : Nat) : Nat := d n + 1
example (a : Nat) : e a = (a + a) + 1 := by

unfold e -- ⊢ d a + 1 = a + a + 1 d is still folded
unfold d -- ⊢ a + a + 1 = a + a + 1
rfl

βδ-conversion

Definition (βδ-conversion, and (con) again)
M=βδ N when the two are linked by a chain of steps, each a β-step or a δ-step,
in either direction.

Δ ;  ⊢ A : B Δ ;  ⊢ B′ : s

Δ ;  ⊢ A : B′
(conv) if B=βδ B′

relation confluent terminating normal form

δ alone yes yes unique; only primitive constants left
β alone yes yes unique
βδ combined yes yes unique

Termination of the combination is a theorem of its own, and it is why rfl
terminates. N&G Def. 9.7.2, 9.7.3, Thm. 6.3.11, 6.3.14, 10.5.2–10.5.8

rfl does β and δ

-- ‘rfl‘ decides βδ-convertibility: it normalises both sides and compares.
def d (n : Nat) : Nat := n + n

example : d 3 = 6 := rfl -- one δ-step, then arithmetic
example : d = fun n => n + n := rfl

-- Three definitions, the third using the other two; unfolding all of them:
def a (x y : Int) : Int := x*x + y*y
def b (x y : Int) : Int := 2 * (x * y)
def c (x y : Int) : Int := a x y + b x y
def lemma’ (x y : Int) : Prop := c x y = (x + y) * (x + y)

example (u v : Int) : lemma’ u v = (u*u + v*v + 2*(u*v) = (u+v)*(u+v)) := rfl
-- Three δ-steps, in whichever order: the normal form is unique.

Unfolding is a policy

-- Which constants the checker unfolds on its own is set per definition.
abbrev Fast : Type := Nat -- @[reducible]: always unfolded
def Norm : Type := Nat -- default: unfolded when needed
@[irreducible] def Slow : Type := Nat -- never unfolded automatically

example : Fast = Nat := rfl
example : Norm = Nat := rfl
#guard_msgs(drop error) in
example : Slow = Nat := rfl -- Type mismatch: rfl has type ?m = ?m

-- but is expected to have type Slow = Nat
unseal Slow in -- the policy, lifted for one declaration
example : Slow = Nat := rfl -- accepted: δ is available again

δ-reduction is one relation on paper; a checker decides convertibility with as few
unfoldings as it can, and each definition says how eagerly it may be opened.

Descriptive and primitive definitions

Definition
 ▷ () := K : L is a descriptive definition;  is a proper constant.
 ▷ () := ⊥⊥ : L is a primitive definition;  is a primitive constant.

⊥⊥ marks the absence of a definiens: a primitive constant cannot be unfolded.
Axioms and primitive notions enter this way:

∅ ▷ N := ⊥⊥ : ∗s a set, postulated
∅ ▷ 0 := ⊥⊥ : N, s := ⊥⊥ : N→N objects, postulated
P:N→∗p ▷ nd(P) := ⊥⊥ : P0⇒ (∀n(Pn⇒ P(sn)))⇒ ∀n(Pn) an axiom scheme
∅ ▷ ET := ⊥⊥ : α: ∗ . (α∨¬α) the excluded third

(deƒ -prm) and (nst-prm) are (deƒ ) and (nst) with the premiss about M
replaced by  ⊢ N : s. N&G Def. 10.2.1, 10.4.2, Ex. 10.2.2, Fig. 10.1

axiom is a definition with no body

-- A primitive definition has no body: ‘axiom‘, and no ‘:=‘.
axiom choiceLike (α : Type) : Nonempty α → α

#print choiceLike
-- axiom choiceLike : (α : Type) → Nonempty α → α
-- no ‘:=‘: nothing to unfold

-- Lean’s own three primitives are of exactly this kind:
#check @propext
#check @Classical.choice
#check @Quot.sound

What survives δ-normalisation

#print axioms opens every box that has a bottom and writes down the ones that do
not; sorry is a bottomless box that fits any type.

N&G Cor. 10.5.4

The audit

-- ‘#print axioms‘: the primitive constants left after δ-normalisation.
theorem plain : 2 + 2 = 4 := rfl
theorem classy (p : Prop) : p ∨ ¬p := Classical.em p

#print axioms plain -- ’plain’ does not depend on any axioms
#print axioms classy
-- ’classy’ depends on axioms: [propext, Classical.choice, Quot.sound]

-- One caveat. A recursor has no body either, yet:
#print axioms Nat.rec -- ’Nat.rec’ does not depend on any axioms
-- The audit lists the constants declared as axioms; the ones every
-- ‘inductive‘ adds, its recursor, it trusts silently.

Why three names, and not one

-- On paper classical logic adds one primitive; the audit reported three.
#print axioms Classical.choice -- depends on axioms: [Classical.choice]
#print axioms propext -- depends on axioms: [propext]
#print axioms Quot.sound -- depends on axioms: [Quot.sound]

-- These two are not: both are theorems, and the audit says what they cost.
#print axioms funext -- depends on axioms: [Quot.sound]
#print axioms Classical.em
-- ’Classical.em’ depends on axioms: [propext, Classical.choice, Quot.sound]

-- ‘funext‘ is proved from quotients; ‘Classical.em‘ from choice, propext
-- and funext, Diaconescu’s theorem. The excluded third is derived.

Diaconescu (1975): choice implies the excluded middle, so Lean derives α∨¬α from its
three primitives. N&G §7.4, Fig. 11.16; TPIL §12.2–12.6

sorry compiles

-- ‘sorry‘ is a primitive constant that inhabits any type: an environment
-- containing it proves everything.
theorem shaky : 2 + 2 = 5 := by sorry

#print axioms shaky -- ’shaky’ depends on axioms: [sorryAx]
#check @sorryAx -- sorryAx : (α : Sort u_1) → Bool → α

-- It compiles, with a warning. Only the audit reveals it.

An axiom is a promise

-- A primitive definition passes every check whether or not it makes sense.
-- k(A,B) := ⊥⊥ : (A ⇒ B) ⇒ A looks like Peirce’s law, and derives False:
axiom bad (A B : Prop) : (A → B) → A

theorem everything : False := bad False False (fun h => h)

#print axioms everything -- ’everything’ depends on axioms: [bad]

-- The kernel is not fooled: it reports the culprit by name. An axiom is
-- a promise, and ‘#print axioms‘ is where the promises are collected.

Logic in λD: the hidden definitions, written down

Last lecture ⊥, ¬ and ∧ were abbreviations; in λD they are descriptive
definitions:

(D1) ⊥() := α: ∗ . α : ∗

A : ∗

(D2) ¬(A) := A→⊥() : ∗

A,B : ∗

(D3) ∧(A,B) := C: ∗ . (A→B→C)→C : ∗

Each line stands for a full derivation, ten lines for D2 alone; the definition
records that its statement is derivable.

The convention: every statement becomes a definition
In a flag derivation, every statement  ⊢ M : N gets a fresh name:
 ▷ c() := M : N.

N&G §11.1–11.3, Figs. 11.1–11.2

The natural-deduction rules, as defined constants

Each natural-deduction rule becomes a definition named after it: its parameter
list is its context, its body the rule’s derivation.

flags definition

A,B:∗p | :A→B ⇒ -n(A,B, ) :=  : A⇒ B
A:∗p | :⊥ ⊥-e(A, ) := A : A
A:∗p | :¬A | :A ¬-e(A, , ) :=  : ⊥
A,B:∗p | :A | :B ∧-n(A,B, , ) := λC: ∗p . λ:A⇒ B⇒ C. : A∧ B
A,B:∗p | :A∧ B ∧-e1(A,B, ) := A (λ:A.λ:B.) : A
A,B:∗p | :A ∨-n1(A,B, ) := λC: ∗p . λ:A⇒ C.λ:B⇒ C.  : A∨ B
A,B,C:∗p | :A∨ B | :A⇒ C |:B⇒ C ∨-e(A,B,C, , ,) := C : C

Two styles for one proof of A⇒ ¬¬A: naming every rule,
3(A) :=⇒ -n(A,¬¬A,λ:A. 2(A, )), or writing the term,
3(A) := λ:A.λ:¬A. . Shorter here; for ∧ and ∨ the names win.

N&G §11.4–11.5, App. A

The named rules are the library

Parameters that only repeat the context may be left out; {a b c : Prop} is that
convention, and @ switches it off. N&G §11.7, Fig. 11.11

The rules, by their library names

-- The library holds the named rules; the parameters that only repeat
-- the context, A B C, are suppressed by ‘{}‘.
#check @And.intro -- @And.intro : ∀ {a b : Prop}, a → b → a ∧ b
#check @Or.elim -- @Or.elim : ∀ {a b c : Prop}, a ∨ b → (a → c) → (b → c) → c
#check @absurd -- @absurd : {a : Prop} → {b : Sort u_1} → a → ¬a → b

variable (A B C : Prop)
example (u : A) (v : B) : A ∧ B := And.intro u v -- ∧-in(A,B,u,v)
example (u : A ∧ B) : A := u.left -- ∧-el1(A,B,u)
example (u : A) : A ∨ B := Or.inl u -- ∨-in1(A,B,u)
example (u : A ∨ B) (v : A → C) (w : B → C) : C :=

Or.elim u v w -- ∨-el(A,B,C,u,v,w)

One line, three meanings

A flag derivation writes each step as a line c() := M : N. Such a line means
three things at once.
1. the statement M : N is derivable in the current environment and context;
2. the definition is appended to Δ, under the name c;
3. every later line may use c() : N.

(1) check M : N (2) record the name (3) use it below

Check, record, use. That is what have does, and it is why have is the most
important tactic in the language. N&G §11.3

have is a flag line

-- ‘have‘ is a flag line: the statement is checked, recorded under a
-- name, and usable below.
example (p q : Prop) (hp : p) (hq : q) : p ∧ q := by
have h1 : p := hp -- checked, recorded, usable below
have h2 : q := hq
exact And.intro h1 h2

The goal state after a have is the flag derivation after that line: one more name in
scope, the goal unchanged.

show, suffices, refine

-- Three more ways to write a flag line.
-- ‘show‘ restates the goal at a βδ-convertible type: (conv), as a tactic.
example (p : Prop) (hp : p) : ¬¬p := by

show ¬p → False
intro hnp; exact hnp hp

-- ‘suffices‘ is ‘have‘ read backwards: name what would be enough, use it
-- at once, and prove it afterwards.
example (p q : Prop) (hp : p) (h : p → q) : q ∧ p := by

suffices hq : q from 〈hq, hp〉
exact h hp

-- ‘refine‘ is ‘exact‘ with holes: each ‘?_‘ becomes a goal, a flag still
-- to be hauled down.
example (p q : Prop) (hp : p) (hq : q) : (p ∧ q) ∧ p := by

refine 〈〈?_, hq〉, ?_〉
· exact hp
· exact hp

A derivation as a list of definitions

Last lecture’s tautology (A∨ B)⇒ (¬A⇒ B), parameters suppressed:

() A,B : ∗p
(b)  : A∨ B
(c) y : ¬A

(d)  : A
(1) 1 := y : ⊥ ¬-e, on (c), (d)
(2) 2 := 1 B : B ⊥-e, on (1)
(3) 3 := λ:A. 2 : A⇒ B ⇒ -n, on (2)
(4) 4 := λ:B. : B⇒ B ⇒ -n
(5) 5 := B3 4 : B ∨-e, on (b), (3), (4)
(6) 6 := λy:¬A. 5 : ¬A⇒ B ⇒ -n, on (5)
(7) 7 := λ:A∨ B. 6 : (A∨ B)⇒ (¬A⇒ B) ⇒ -n, on (6)

Seven lines, seven names; unfolding (7) down to 1 gives last lecture’s term.
N&G §11.6–11.7

The script is the derivation, written from the bottom

N&G Fig. 11.15; TPIL §5.4

Classical logic in λD

The excluded third is one primitive definition in the environment:

A : ∗p

(1) ec-thrd(A) := ⊥⊥ : A∨¬A primitive

(2) 2(A) := λ:A. : A⇒ A

 : ¬¬A

 : ¬A

(3) 3(A, , ) :=  : ⊥ (4) 4(A, , ) := 3(A, , )A : A

(5) 5(A, ) := λ:¬A. 4(A, , ) : ¬A⇒ A

(6) 6(A, ) := ec-thrd(A)A2(A)5(A, ) : A ∨-e

(7) dob-neg(A) := λ:¬¬A. 6(A, ) : ¬¬A⇒ A

From dob-neg comes a strategy: to prove A, raise a flag ¬A and derive ⊥.
N&G §11.8, Rem. 11.8.1

The case split, and what the audit says of it

-- On paper exc-thrd(A) := ⊥⊥ : A ∨ ¬A is primitive; in Lean it is a
-- theorem, and a script that uses it is classical:
theorem doubNeg (A : Prop) (u : ¬¬A) : A := by

by_cases h : A -- exc-thrd(A), as a case split
· exact h
· exact (u h).elim -- ¬¬A and ¬A give ⊥, and ⊥-el

theorem doubNeg’ (A : Prop) (u : ¬¬A) : A :=
Classical.byContradiction u -- assume ¬A, derive ⊥

theorem constructive (A : Prop) : A → ¬¬A := fun a f => f a

#print axioms doubNeg -- depends on axioms: [propext, Classical.choice, Quot.sound]
#print axioms constructive -- ’constructive’ does not depend on any axioms

by_cases is ∨-elimination on ec-thrd(A); Classical.byContradiction is line (7) by
name. Both are classical, and #print axioms says so; the constructive direction reports
nothing.

∃-elimination raises two flags

To use ∃:S. P while proving A, one cannot extract a witness; the rule wants
∀:S. (P⇒ A), built by raising two flags:

(. . . ) := · · · : ∃:S. P
 : S

 : P

b(. . . , , ) := . . . : A

c(. . . , ) := λ:P. b(. . . ) : P⇒ A

d(. . . ) := λ:S. c(. . . ) : ∀:S. (P⇒ A)

e(. . . ) := ∃-e(S, P, , A, d) : A

The goal A is unchanged; only the context is larger. That is obtain. Side condition: 
not free in A. N&G §11.10, Rem. 11.10.2

obtain is ∃-elimination

-- To use ∃x:S. P x while proving A, raise two flags, x : S and u : P x,
-- prove A under them, and ∃-el discharges both. That is ‘obtain‘.
example (S : Type) (P : S → Prop) (Q : Prop)

(h : ∃ x, P x) (k : ∀ x, P x → Q) : Q := by
obtain 〈x, hx〉 := h -- raises the flags x : S and hx : P x
exact k x hx -- prove Q under them

-- ∃x.Px ⇒ ∀y.(Py ⇒ Qy) ⇒ ∃z.Qz, a six-line derivation in two:
example (S : Type) (P Q : S → Prop) (u : ∃ x, P x) (v : ∀ y, P y → Q y) :

∃ z, Q z := by
obtain 〈x, w〉 := u -- the flags x : S, w : P x
exact 〈x, v x w〉 -- (1) v x, (2) a1 w, (3) ∃-in, (5) ∃-el

-- The side condition: x must not occur free in the goal. Lean enforces
-- it: the goal may not mention ‘x‘.

Devised backwards, recorded forwards

A tactic script is the record of a search whose successful branch became a term; the
failed attempts are not recorded. N&G Rem. 11.9.3

The script is not the term

-- A derivation is checked from top to bottom, but devised the other way:
-- from the goal, with the failed attempts left unrecorded.
theorem demorgan (A B : Prop) (h : ¬A ∨ ¬B) : ¬(A ∧ B) := by

intro hab -- backwards: from the goal
cases h with
| inl ha => exact ha hab.left
| inr hb => exact hb hab.right

#print demorgan -- forwards: only the result survives
-- fun A B h hab =>
-- Or.casesOn (motive := fun t => h = t → False) h
-- (fun ha h => ha hab.left) (fun hb h => hb hab.right) (Eq.refl h)
-- The ‘cases‘ became a recursor application. The script is not the term.

Search in the environment: exact?

example (a b : Nat) : a + b = b + a := by exact? -- Try this: exact Nat.add_comm a b
example (p q : Prop) (hp : p) (hq : q) : p ∧ q := by exact? -- Try this: exact 〈hp, hq〉

Why is a Lean proof trusted?

1 a term is a complete coding of its proof N&G Rem. 2.9.1
2 Type Checking is decidable, from λ→ up to λD N&G Thm. 2.10.10, 6.3.15, 10.5.8
3 the de Bruijn criterion: any means may produce the

term, as long as a small kernel checks it
N&G §16.2

4 the primitive constants a term rests on can be listed N&G Cor. 10.5.4

The de Bruijn criterion

informal proof p informal statement A

formal proof p formal statement A

find type find type

p ok, type is T A ok

check T = A

p has type A: p proves A

failure → report

Only the dashed arrows are not done by the computer; the final check is
βδ-convertibility. The de Bruijn criterion: a proof assistant generates proof terms a
small kernel can check, whatever produced them. N&G §16.2, Fig. 16.1

What you are trusting: the trusted computing base

Computational Logic in the Era of AI 49 / 53



A first attempt at mathematics, and what it lacks

Lemma: a least element of a partially ordered set is unique. Its proof, as a flag
derivation:

S : ∗s
m1,m2 : S

 : ∀n∈S(m1 ≤ n)
 : ∀n∈S(m2 ≤ n)

(1) 1 := t1 : m1 ≤m2 (2) 2 := t2 :m2 ≤m1

(3) 3 := t3 : m1=m2

(4) 4 := t4 : the least element is unique

not yet in Δ what must be defined

the symbol = in line (3) equality, as a notion in λD
the term t3 how two ≤ give an =
“S has a least element” sets and predicates, with ∃
“unique” ∃!, from = and ∃
the term t4 a proof using all of the above

The programme
Every gap is a definition still missing
from Δ. Equality and sets come next
lecture; numbers and induction after
that.

N&G §12.1, Fig. 12.1

